
Motivation

Recall: instruction memory is a flat list (with the program counter as index)!  

if/while/for/break/continue

We don’t get to keep 

Instruction pointer 
(a.k.a. program counter) 

register holds address of next instruction to execute

%rip

Address Contents
0x… <instruction>
0x… <instruction>
0x… <instruction>
0x… <instruction>
0x… <instruction>
…



Conditionals and Control Flow

Two key pieces 
1. Comparisons and tests: check conditions 
2. Transfer control: choose next instruction 

To implement familiar C constructs 
● if else
● while
● do while
● for
● break
● continue

Instruction pointer 
(a.k.a. program counter) 

register holds address of next instruction to execute

Condition codes (a.k.a. flags) 
1-bit registers hold flags set by last ALU operation 

	 	 Zero Flag	 	 result == 0 

	 	 Sign Flag	 	 result < 0 

	 	 Carry Flag	 	 carry-out/unsigned overflow 

	 	 Overflow Flag	 two's complement overflow

%rip

CF

ZF

SF

OF

Processor Control-Flow State



1. Compare and test: conditions
cmpq b,a computes a - b,	 sets flags, discards result

	 	 Which flags indicate that a < b ?  (signed? unsigned?) 

testq b,a computes a & b,	 sets flags, discards result

	 	 Common pattern: 

testq %rax, %rax
What do ZF and  SF indicate?



2. Transfer control: choose next instruction
Different jump/branch instructions to  different part of code by setting %rip.

j__ Condition Description

Conditional 
jumps

je ZF Equal / Zero

jne ~ZF Not Equal / Not Zero

js SF Negative

jns ~SF Nonnegative

jg ~(SF^OF)&~ZF Greater (Signed)

jge ~(SF^OF) Greater or Equal (Signed)

jl (SF^OF) Less (Signed)

jle (SF^OF)|ZF Less or Equal (Signed)

ja ~CF&~ZF Above (unsigned)

jb CF Below (unsigned)

Unconditional 
jump jmp 1 Unconditional



Jump for control flow

cmpq %rax,%rcx
je label
…
…
…
addq %rdx,%raxlabel:

Label 
Name for address of 

following item.

Jump immediately follows comparison/test. 
Together, they make a decision: 

"if %rcx == %rax  then jump to label."

Executed only if 
%rax ≠ %rcx



Interpreting Conditional Jumps
It is easier to read conditional jumps in x86-64 by comparing b 
against a instead of looking at condition codes.

cmp b,a test b,a

je “Equal” a == b a&b == 0

jne “Not equal” a != b a&b != 0

js “Sign” (negative) a-b < 0 a&b <  0

jns (non-negative) a-b >= 0 a&b >= 0

jg “Greater” a > b a&b >  0

jge “Greater or equal” a >= b a&b >= 0

jl “Less” a < b a&b <  0

jle ”Less or equal” a <= b a&b <= 0

ja “Above” (unsigned >) a > b a&b > 0U

jb “Below” (unsigned <) a < b a&b < 0U

    cmpq 5, (p) 
je:  *p == 5 . 
jne: *p != 5 . 
jg:  *p >  5 . 
jl:  *p <  5 .

   testq a, a 
je:   a == 0 . 
jne:  a != 0 . 
jg:   a >  0 . 
jl:   a <  0 .



Conditional branch example

long absdiff(long x,long y) {
  long result;
  if (x > y) {
    result = x-y;
  } else {
    result = y-x;
  }
  return result;
}

How did the compiler create this?

Labels 
Name for address of 

following item.

absdiff:
cmpq   %rsi, %rdi
jle    .L7
subq   %rsi, %rdi
movq   %rdi, %rax

.L8:
retq

.L7:
subq   %rdi, %rsi
movq   %rsi, %rax
jmp    .L8



Nodes = Basic Blocks: 
Straight-line code always 
executed together in order.

Control-Flow Graph

long absdiff(long x, long y){
  long result;
  if (x > y) {
    result = x-y;
  } else {
    result = y-x;
  }
  return result;
}

long result;
if (x > y)  else  

result = x-y; result = y-x;

return result;

Code flowchart/directed graph.

Introduced by Fran Allen, et al. 
Won the 2006 Turing Award 

for her work on compilers.

Edges = Control Flow: 
Which basic block executes 
next (under what condition).

then else

long result;
if (x > y)  else  

result = x-y; result = y-x;

return result;

Control-Flow Graph

How do we represent this non-flat structure in a single instruction memory? 



Choose a linear order of basic blocks.

long result;
if (!(x > y)) 

result = x-y;

result = y-x;

return result;

then else



Translate basic blocks with jumps + labels

cmpq   %rsi, %rdi
jle    Else

subq   %rsi, %rdi
movq   %rdi, %rax

subq   %rdi, %rsi
movq   %rsi, %rax
jmp End

retq

Else:

End:

long result;
if (!(x > y)) 

result = x-y;

result = y-x;

return result;



Execute absdiff

%rax

%rdi

%rsi

5

3

Registers

cmpq   %rsi, %rdi
jle    Else

subq   %rsi, %rdi
movq   %rdi, %rax

retq

Else:

End:

subq   %rdi, %rsi
movq   %rsi, %rax
jmp End

ex



Execute absdiff

%rax

%rdi

%rsi

2

5 2

3

Registers

cmpq   %rsi, %rdi
jle    Else

subq   %rsi, %rdi
movq   %rdi, %rax

retq

Else:

End:

subq   %rdi, %rsi
movq   %rsi, %rax
jmp End

ex



Execute absdiff

%rax

%rdi

%rsi

2

5 2

3

Registers

cmpq   %rsi, %rdi
jle    Else

subq   %rsi, %rdi
movq   %rdi, %rax

retq

Else:

End:

subq   %rdi, %rsi
movq   %rsi, %rax
jmp End

ex



Execute absdiff

%rax

%rdi

%rsi

4

7

Registers

cmpq   %rsi, %rdi
jle    Else

subq   %rsi, %rdi
movq   %rdi, %rax

retq

Else:

End:

subq   %rdi, %rsi
movq   %rsi, %rax
jmp End

ex



Execute absdiff

%rax

%rdi

%rsi

3

4

7 3

Registers

cmpq   %rsi, %rdi
jle    Else

subq   %rsi, %rdi
movq   %rdi, %rax

retq

Else:

End:

subq   %rdi, %rsi
movq   %rsi, %rax
jmp End

ex



Execute absdiff

18

%rax

%rdi

%rsi

3

4

7 3

Registers

cmpq   %rsi, %rdi
jle    Else

subq   %rsi, %rdi
movq   %rdi, %rax

retq

Else:

End:

subq   %rdi, %rsi
movq   %rsi, %rax
jmp End

ex



Note: CSAPP shows translation with goto

long goto_ad(long x,long y){
  int result;
  if (x <= y) goto Else;
  result = x-y;
End:
  return result;
Else:
  result = y-x;
  goto End;
}

long absdiff(long x,long y){
  int result;
  if (x > y) {
    result = x-y;
  } else {
    result = y-x;
  }
  return result;
}

http://xkcd.com/292/

But never use goto in your source code!



Compile if-else

long wacky(long x, long y){
  long result;
  if (x + y > 7) {
    result = x;
  } else {
    result = y + 2;
  }
  return result;
}

Recall:  
x is available in %rdi 
y is available in %rsi
result should be in %rax for return 

Instructions to use:  
movq, addq, cmpq, jle or jg

wacky:

ex



Compile if-else (solution #1)

long wacky(long x, long y){
  long result;
  if (x + y > 7) {
    result = x;
  } else {
    result = y + 2;
  }
  return result;
}

wacky:
  movq %rdi, %rdx
  addq %rsi, %rdx
  cmpq $7, %rdx
  jle Else

  movq %rdi, %rax

 End:
  retq

 Else:
  addq $2, %rsi
  movq %rsi, %rax
  jmp End

ex

Recall:  
x is available in %rdi 
y is available in %rsi
result should be in %rax for return



Compile if-else (solution #2) - leaq

long wacky(long x, long y){
  long result;
  if (x + y > 7) {
    result = x;
  } else {
    result = y + 2;
  }
  return result;
}

wacky:
  leaq (%rdi, %rsi), %rdx
  cmpq $7, %rdx
  jle Else

  movq %rdi, %rax

 End:
  retq

 Else:
  leaq 2(%rsi), %rax
  jmp End

ex

Recall:  
x is available in %rdi 
y is available in %rsi
result should be in %rax for return



Encoding jumps: PC-relative addressing

PC-relative offsets support relocatable code. 
Absolute branches do not (or it's hard).

0x100     cmpq  %rax, %rbx    0x1000  
0x102     je   0x70           0x1002  
0x104     …                   0x1004  
…         …                   …  
0x174     addq  %rax, %rbx    0x1074  
long fact_do(long x) {
  // Assume x >= 1
  long result = 1;
  do {
    result = result * x;
    x = x - 1;
  } while (x > 1);
  return result;
}

do while  loop

long result = 1;

result = result*x;
x = x - 1;

(x > 1) ?

return result;

Yes
No



do while  loop

fact_do:
movq $1,%rax

.L11:
imulq %rdi,%rax
decq %rdi

cmpq $1,%rdi
jg .L11

retq

Why put the loop condition at the end?

long result = 1;

result = result*x;
x = x - 1;

(x > 1) ?

return result;

Yes
No



long fact_while(long x){  
  // Assume x >= 0
  long result = 1;
  while (x > 1) {
    result = result * x;
    x = x - 1;
  }
  return result;
}

while  loop

long result = 1;

result = result*x;
x = x - 1;

(x > 1) ?

return result;
This order is used by GCC for x86-64. Why?

long result = 1;

result = result*x;
x = x - 1;

(x > 1) ?

return result;

Yes
No

Yes

No



long fact_while(long x){
  // Assume x >= 0
  long result = 1;
  while (x > 1) {
    result = result * x;
    x = x - 1;
  }
  return result;
} fact_while:

  movq $1, %rax
  jmp   .L34

.L35:
  imulq %rdi, %rax
  decq  %rdi

.L34:
  cmpq  $1, %rdi
  jg    .L35

  retq

while  loop

int result = 1;

result = result * x;
x = x - 1;

(x > 1) ?

return result;

Yes
No



for  loop translation

for (Initialize; Test; Update ) {  
Body 

}

for (result = 1; p != 0; p = p>>1) {
  if (p & 0x1) {
    result = result * x;
  }
  x = x * x;
}

result = 1;

if (p & 0x1) {
  result = result*x;
}
x = x * x;
p = p >> 1;

(p != 0) ? Yes
No

Initialize;
while (Test) {
    Body ; 
    Update; 
}

for loops are syntactic 
sugar for while loops: 
we can just translate for 
to while

Initialize

Body
Update

Test ? Yes
No



for loop: square-and-multiply

Algorithm 
Exploit bit representation: p = p0 + 2p1 + 22p2 + … 2n–1pn–1 

Gives: xp = z0 · z1 2 · (z2 2) 2 · … · (…((zn –12) 2 )…) 2 
zi = 1  when pi = 0
zi = x  when pi = 1

Complexity O(log p) = O(sizeof(p)) 31

/* Compute x raised to nonnegative power p */
int power(int x, unsigned int p) {

int result;
for (result = 1; p != 0; p = p>>1) {

if (p & 0x1) {
result = result * x;

     }
x = x*x;

  }
return result;

}

n–1  times

Example 

311	 = 31 * 32 * 38 

	 = 31 * 32 * ((32)2)2 

                                               xm * xn = xm+n 
0          ...    0       1      0     1      1   = 11 
12^31 * ... * 116 * x8 * 14 * x2 * x1 = x11 

1 = x0    x = x1 

optional



for loop: power iterations

/* Compute x raised to nonnegative power p */
int power(int x, unsigned int p) {

int result;
for (result = 1; p != 0; p = p>>1) {

if (p & 0x1) {
result = result * x;

     }
x = x*x;

}
return result;

} iteration
s

result x p
0 1 3 11 = 10112
1 3 9 5 =  1012
2 27 81 2 =   102
3 27 6561 1 =    12
4 177147 430467

21
02

optional



(Aside) Conditional Move

absdiff:
  movq    %rdi, %rax 
  subq    %rsi, %rax 
  movq    %rsi, %rdx 
  subq    %rdi, %rdx 
  cmpq    %rsi, %rdi 
  cmovle  %rdx, %rax 
  ret

long absdiff(long x, long y) {
  return x>y ? x-y : y-x;
}

long absdiff(long x,long y) {
  long result;
  if (x > y) {
    result = x - y;
  } else {
    result = y - x;
  }
  return result;
}

Why?  Branch prediction in pipelined/OoO processors.

cmov_ src, dest

if (Test) Dest     Src←



Expensive Computations

(Aside) Bad uses of conditional move

 val = Test(x) ? Hard1(x) : Hard2(x);

Risky Computations

 val = p ? *p : 0;

Computations with side effects

 val = x > 0 ? x++ : x--;



long switch_eg (long x, long y, long z) {
    long w = 1;
    switch(x) {
    case 1:
        w = y * z;
        break;
    case 2:
        w = y - z;
    case 3:
        w += z;
        break;
    case 5:
    case 6:
        w -= z;
        break;
    default:
        w = 2;
    }
    return w;
}

switch  statement

35

Fall through cases

Multiple case labels

Missing cases use default

Lots to manage:  
   use a jump table.



switch jump table structure

switch(x) {
  case 1: <some code>
          break;
  case 2: <some code>
  case 3: <some code>
          break;
  case 5:
  case 6: <some code>
          break;
  default: <some code>
}

6
5
4
3
2

Jump 
Table

Code 
Blocks

Memory

Translation sketch:

if (0 <= x && x <= 6)
  addr = jumptable[x];
  goto addr;
else
  goto default;

C code:

1
0

Each row in the jump table is the 
address of the code for that case



switch jump table assembly declaration

.section .rodata
   .align 8
.L4:
.quad   .L8  # x == 0
.quad   .L3  # x == 1
.quad   .L5  # x == 2
.quad   .L9  # x == 3
.quad   .L8  # x == 4
.quad   .L7  # x == 5
.quad   .L7  # x == 6

    switch(x) {
    case 1:      // .L3
        w = y * z;
        break;
    case 2:      // .L5
        w = y - z;

    case 3:      // .L9
        w += z;
        break;
    case 5:
    case 6:      // .L7
        w -= z;
        break;
    default:     // .L8
        w = 2;
    }

“quad” = q suffix = 8-byte value

read-only data 
(not instructions)

8-byte 
alignment



switch case dispatch

switch_eg:
movl  $1, %eax
cmpq  $6, %rdi
ja    .L8
jmp   *.L4(,%rdi,8)

long switch_eg(long x, long y, long z) {
    long w = 1;
    switch(x) {
      . . .
    }
    return w;
}

Jump table

.section .rodata
   .align 8
.L4:
.quad   .L8  # x == 0
.quad   .L3  # x == 1
.quad   .L5  # x == 2
.quad   .L9  # x == 3
.quad   .L8  # x == 4
.quad   .L7  # x == 5
.quad   .L7  # x == 6

Jump if above (unsigned, but…) 

indirect jump



switch cases

.L3:  movq    %rsi, %rax
      imulq   %rdx, %rax
      retq
.L5:  movq    %rsi, %rax
      subq    %rdx, %rax
.L9:
      addq    %rdx, %rax
      retq

.L7:
      subq  %rdx, %rax
      retq
.L8:
      movl  $2, %eax
      retq

  switch(x) {
  case 1:    // .L3
      w = y * z;
      break;
  case 2:    // .L5
      w = y - z;
  case 3:    // .L9
      w += z;
      break;
  case 5:    // .L7
  case 6:    // .L7
      w -= z;
      break;
  default:   // .L8
      w = 2;  
  }
  return w;

“inlined" return

Fall-through

Aside: movl is used because 2 is a small positive value that fits  
in 32 bits. High order bits of %rax get set to zero automatically.  
It takes fewer bytes to encode a literal  movl vs a movq.

Reg. Use

%rdi x
%rsi y
%rdx z
%rax w



switch machine code

00000000004004f6 <switch_eg>:
  . . .
  4004fd:  77 2b                   ja 40052a <switch_eg+0x34>
  4004ff:  ff 24 fd d0 05 40 00    jmpq *0x4005d0(,%rdi,8)

switch_eg:
  . . .
  cmpq  $6, %rdi
  ja    .L8
  jmp   *.L4(,%rdi,8)

Assembly Code

Disassembled Object Code

Inspect jump table contents using GDB. 
Examine contents as 7 addresses 

(gdb) x/7a 0x4005d0 
0x4005d0:  0x40052a <switch_eg+52>    0x400506 <switch_eg+16> 
0x4005e0:  0x40050e <switch_eg+24>    0x400518 <switch_eg+34> 
0x4005f0:  0x40052a <switch_eg+52>    0x400521 <switch_eg+43> 
0x400600:  0x400521 <switch_eg+43>

Address of code for case 0 Address of code for case 1

Address of code for case 6

When looking as disassembled 
code: an indirect jump like this 
is a sign it’s a jump table 
encoding a switch 



Would you implement this with a jump table?

switch(x) {
  case 0:     <some code>
              break;
  case 10:    <some code>
              break;
  case 52000: <some code>
              break;
  default:    <some code>
              break;
}

ex


